Rules for integrands of the form $\text{Trig}[d+e\,x]^m\,(a+b\,\text{Cos}[d+e\,x]^p+c\,\text{Sin}[d+e\,x]^q)^n$

1. $\int \text{Sin}[d+e\,x]^m\,(a+b\,\text{Cos}[d+e\,x]^p+c\,\text{Sin}[d+e\,x]^q)^n\,dx$ when $\frac{m}{2}\in\mathbb{Z}\land\frac{p}{2}\in\mathbb{Z}\land\frac{q}{2}\in\mathbb{Z}\land n\in\mathbb{Z}$

1: $\int \text{Sin}[d+e\,x]^m\,(a+b\,\text{Cos}[d+e\,x]^p+c\,\text{Sin}[d+e\,x]^q)^n\,dx$ when $\frac{m}{2}\in\mathbb{Z}\land\frac{p}{2}\in\mathbb{Z}\land\frac{q}{2}\in\mathbb{Z}\land n\in\mathbb{Z}\land 0<p\le q$

Derivation: Integration by substitution

Basis: $\text{Cos}[z]^2 == \frac{\text{Cot}[z]^2}{1+\text{Cot}[z]^2}$

Basis: $\text{Sin}[z]^2 == \frac{1}{1+\text{Cot}[z]^2}$

Basis: If $\frac{m}{2}\in\mathbb{Z}$, then

$$\text{Sin}[d+e\,x]^m\,F\left[\text{Cos}[d+e\,x]^2,\,\text{Sin}[d+e\,x]^2\right] == -\frac{1}{e}\,\text{Subst}\left[\frac{F\left[\frac{x^2}{1+x^2},\frac{1}{1+x^2}\right]}{(1+x^2)^{m/2+1}},\,x,\,\text{Cot}[d+e\,x]\right]\partial_x\text{Cot}[d+e\,x]$$

Rule: If $\frac{m}{2}\in\mathbb{Z}\land\frac{p}{2}\in\mathbb{Z}\land\frac{q}{2}\in\mathbb{Z}\land n\in\mathbb{Z}\land 0<p\le q$, then

$$\int \text{Sin}[d+e\,x]^m\,(a+b\,\text{Cos}[d+e\,x]^p+c\,\text{Sin}[d+e\,x]^q)^n\,dx \to -\frac{1}{e}\,\text{Subst}\left[\int\frac{\left(c+b\,x^p\,(1+x^2)^{\frac{q}{2}-\frac{p}{2}}+a\,(1+x^2)^{q/2}\right)^n}{(1+x^2)^{m/2+n\,q/2+1}}\,dx,\,x,\,\text{Cot}[d+e\,x]\right]$$

Program code:

```
Int[sin[d_.+e_.*x_]^m_*(a_+b_.*cos[d_.+e_.*x_]^p_+c_.*sin[d_.+e_.*x_]^q_)^n_,x_Symbol] :=
  Module[{f=FreeFactors[Cot[d+e*x],x]},
  -f/e*Subst[Int[ExpandToSum[c+b*(1+f^2*x^2)^(q/2-p/2)+a*(1+f^2*x^2)^(q/2),x]^n/(1+f^2*x^2)^(m/2+n*q/2+1),x],x,Cot[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[m/2] && IntegerQ[p/2] && IntegerQ[q/2] && IntegerQ[n] && GtQ[p,0] && LeQ[p,q]
```

```
Int[cos[d_.+e_.*x_]^m_*(a_+b_.*sin[d_.+e_.*x_]^p_+c_.*cos[d_.+e_.*x_]^q_)^n_,x_Symbol] :=
  Module[{f=FreeFactors[Tan[d+e*x],x]},
  f/e*Subst[Int[ExpandToSum[c+b*(1+f^2*x^2)^(q/2-p/2)+a*(1+f^2*x^2)^(q/2),x]^n/(1+f^2*x^2)^(m/2+n*q/2+1),x],x,Tan[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[m/2] && IntegerQ[p/2] && IntegerQ[q/2] && IntegerQ[n] && GtQ[p,0] && LeQ[p,q]
```

2: $\int \sin[d+e\,x]^m \left(a+b\cos[d+e\,x]^p+c\sin[d+e\,x]^q\right)^n dx$ when $\frac{m}{2}\in\mathbb{Z}\land\frac{p}{2}\in\mathbb{Z}\land\frac{q}{2}\in\mathbb{Z}\land n\in\mathbb{Z}\land 0<q<p$

Derivation: Integration by substitution

Basis: $\cos[z]^2 == \frac{\cot[z]^2}{1+\cot[z]^2}$

Basis: $\sin[z]^2 == \frac{1}{1+\cot[z]^2}$

Basis: If $\frac{m}{2}\in\mathbb{Z}$, then

$$\sin[d+e\,x]^m F\left[\cos[d+e\,x]^2,\ \sin[d+e\,x]^2\right] == -\frac{1}{e}\,\text{Subst}\left[\frac{F\left[\frac{x^2}{1+x^2},\frac{1}{1+x^2}\right]}{\left(1+x^2\right)^{m/2+1}},\ x,\ \cot[d+e\,x]\right]\partial_x \cot[d+e\,x]$$

Rule: If $\frac{m}{2}\in\mathbb{Z}\land\frac{p}{2}\in\mathbb{Z}\land\frac{q}{2}\in\mathbb{Z}\land n\in\mathbb{Z}\land 0<q<p$, then

$$\int \sin[d+e\,x]^m \left(a+b\cos[d+e\,x]^p+c\sin[d+e\,x]^q\right)^n dx \to -\frac{1}{e}\,\text{Subst}\left[\int\frac{\left(a\left(1+x^2\right)^{p/2}+b\,x^p+c\left(1+x^2\right)^{\frac{p}{2}-\frac{q}{2}}\right)^n}{\left(1+x^2\right)^{m/2+n\,p/2+1}}dx,\ x,\ \cot[d+e\,x]\right]$$

–

Program code:

```
Int[sin[d_.+e_.*x_]^m_*(a_+b_.*cos[d_.+e_.*x_]^p_+c_.*sin[d_.+e_.*x_]^q_)^n_,x_Symbol] :=
  Module[{f=FreeFactors[Cot[d+e*x],x]},
  -f/e*Subst[Int[ExpandToSum[a*(1+f^2*x^2)^(p/2)+b*f^p*x^p+c*(1+f^2*x^2)^(p/2-q/2),x]^n/(1+f^2*x^2)^(m/2+n*p/2+1),x],x,
    Cot[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[m/2] && IntegerQ[p/2] && IntegerQ[q/2] && IntegerQ[n] && LtQ[0,q,p]
```

```
Int[cos[d_.+e_.*x_]^m_*(a_+b_.*sin[d_.+e_.*x_]^p_+c_.*cos[d_.+e_.*x_]^q_)^n_,x_Symbol] :=
  Module[{f=FreeFactors[Tan[d+e*x],x]},
  f/e*Subst[Int[ExpandToSum[a*(1+f^2*x^2)^(p/2)+b*f^p*x^p+c*(1+f^2*x^2)^(p/2-q/2),x]^n/(1+f^2*x^2)^(m/2+n*p/2+1),x],x,
    Tan[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[m/2] && IntegerQ[p/2] && IntegerQ[q/2] && IntegerQ[n] && LtQ[0,q,p]
```

2. $\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Cos}[d+e\,x]^p+c\,\text{Sin}[d+e\,x]^q\right)^n dx$ when $\frac{m}{2}\in\mathbb{Z}\land\frac{p}{2}\in\mathbb{Z}\land\frac{q}{2}\in\mathbb{Z}\land n\in\mathbb{Z}$

1: $\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Cos}[d+e\,x]^p+c\,\text{Sin}[d+e\,x]^q\right)^n dx$ when $\frac{m}{2}\in\mathbb{Z}\land\frac{p}{2}\in\mathbb{Z}\land\frac{q}{2}\in\mathbb{Z}\land n\in\mathbb{Z}\land 0<p\le q$

Derivation: Integration by substitution

Basis: $\text{Cos}[z]^2 == \frac{\text{Cot}[z]^2}{1+\text{Cot}[z]^2}$

Basis: $\text{Sin}[z]^2 == \frac{1}{1+\text{Cot}[z]^2}$

Basis: If $\frac{m}{2}\in\mathbb{Z}$, then

$$\text{Cos}[d+e\,x]^m\,F\left[\text{Cos}[d+e\,x]^2,\,\text{Sin}[d+e\,x]^2\right] == -\frac{1}{e}\,\text{Subst}\left[\frac{F\left[\frac{x^2}{1+x^2},\frac{1}{1+x^2}\right]}{\left(1+x^2\right)^{m/2+1}},\,x,\,\text{Cot}[d+e\,x]\right]\partial_x\,\text{Cot}[d+e\,x]$$

Rule: If $\frac{m}{2}\in\mathbb{Z}\land\frac{p}{2}\in\mathbb{Z}\land\frac{q}{2}\in\mathbb{Z}\land n\in\mathbb{Z}\land 0<p\le q$, then

$$\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Cos}[d+e\,x]^p+c\,\text{Sin}[d+e\,x]^q\right)^n dx \to -\frac{1}{e}\,\text{Subst}\left[\int\frac{\left(c+b\,x^p\left(1+x^2\right)^{\frac{q}{2}-\frac{p}{2}}+a\left(1+x^2\right)^{q/2}\right)^n}{\left(1+x^2\right)^{m/2+n\,q/2+1}}\,dx,\,x,\,\text{Cot}[d+e\,x]\right]$$

Program code:

```
Int[sin[d_.+e_.*x_]^m_*(a_+b_.*cos[d_.+e_.*x_]^p_+c_.*sin[d_.+e_.*x_]^q_)^n_,x_Symbol] :=
  Module[{f=FreeFactors[Cot[d+e*x],x]},
  -f/e*Subst[Int[ExpandToSum[c+b*(1+f^2*x^2)^(q/2-p/2)+a*(1+f^2*x^2)^(q/2),x]^n/(1+f^2*x^2)^(m/2+n*q/2+1),x],x,Cot[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[m/2] && IntegerQ[p/2] && IntegerQ[q/2] && IntegerQ[n] && GtQ[p,0] && LeQ[p,q]
```

```
Int[cos[d_.+e_.*x_]^m_*(a_+b_.*sin[d_.+e_.*x_]^p_+c_.*cos[d_.+e_.*x_]^q_)^n_,x_Symbol] :=
  Module[{f=FreeFactors[Tan[d+e*x],x]},
  f/e*Subst[Int[ExpandToSum[c+b*(1+f^2*x^2)^(q/2-p/2)+a*(1+f^2*x^2)^(q/2),x]^n/(1+f^2*x^2)^(m/2+n*q/2+1),x],x,Tan[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[m/2] && IntegerQ[p/2] && IntegerQ[q/2] && IntegerQ[n] && GtQ[p,0] && LeQ[p,q]
```

2: $\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Cos}[d+e\,x]^p+c\,\text{Sin}[d+e\,x]^q\right)^n dx$ when $\frac{m}{2}\in\mathbb{Z}\land\frac{p}{2}\in\mathbb{Z}\land\frac{q}{2}\in\mathbb{Z}\land n\in\mathbb{Z}\land 0<q<p$

Derivation: Integration by substitution

Basis: $\text{Cos}[z]^2 = \frac{\text{Cot}[z]^2}{1+\text{Cot}[z]^2}$

Basis: $\text{Sin}[z]^2 = \frac{1}{1+\text{Cot}[z]^2}$

Basis: If $\frac{m}{2}\in\mathbb{Z}$, then

$$\text{Sin}[d+e\,x]^m F\left[\text{Cos}[d+e\,x]^2,\ \text{Sin}[d+e\,x]^2\right] = -\frac{1}{e}\,\text{Subst}\left[\frac{F\left[\frac{x^2}{1+x^2},\frac{1}{1+x^2}\right]}{\left(1+x^2\right)^{m/2+1}},\ x,\ \text{Cot}[d+e\,x]\right]\partial_x\text{Cot}[d+e\,x]$$

Rule: If $\frac{m}{2}\in\mathbb{Z}\land\frac{p}{2}\in\mathbb{Z}\land\frac{q}{2}\in\mathbb{Z}\land n\in\mathbb{Z}\land 0<q<p$, then

$$\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Cos}[d+e\,x]^p+c\,\text{Sin}[d+e\,x]^q\right)^n dx \to -\frac{1}{e}\,\text{Subst}\left[\int\frac{\left(a\left(1+x^2\right)^{p/2}+b\,x^p+c\left(1+x^2\right)^{\frac{p}{2}-\frac{q}{2}}\right)^n}{\left(1+x^2\right)^{m/2+n\,p/2+1}}dx,\ x,\ \text{Cot}[d+e\,x]\right]$$

–

Program code:

```
Int[sin[d_.+e_.*x_]^m_*(a_+b_.*cos[d_.+e_.*x_]^p_+c_.*sin[d_.+e_.*x_]^q_)^n_,x_Symbol] :=
  Module[{f=FreeFactors[Cot[d+e*x],x]},
  -f/e*Subst[Int[ExpandToSum[a*(1+f^2*x^2)^(p/2)+b*f^p*x^p+c*(1+f^2*x^2)^(p/2-q/2),x]^n/(1+f^2*x^2)^(m/2+n*p/2+1),x],x,
    Cot[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[m/2] && IntegerQ[p/2] && IntegerQ[q/2] && IntegerQ[n] && LtQ[0,q,p]
```

```
Int[cos[d_.+e_.*x_]^m_*(a_+b_.*sin[d_.+e_.*x_]^p_+c_.*cos[d_.+e_.*x_]^q_)^n_,x_Symbol] :=
  Module[{f=FreeFactors[Tan[d+e*x],x]},
  f/e*Subst[Int[ExpandToSum[a*(1+f^2*x^2)^(p/2)+b*f^p*x^p+c*(1+f^2*x^2)^(p/2-q/2),x]^n/(1+f^2*x^2)^(m/2+n*p/2+1),x],x,
    Tan[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[m/2] && IntegerQ[p/2] && IntegerQ[q/2] && IntegerQ[n] && LtQ[0,q,p]
```